


RETURN TO
SCIENTIFIC & TECHNICAL INFORMATION DIVISION
(ESTI), BUILDING 1211

ESD ACCESSION LIST

ESTI Call No. 66427

Copy No. of cys.

ESTI FILE COPY

<p style="text-align: center;">Technical Note</p>	<p style="text-align: center;">1969-42</p>
<p style="text-align: center;">Some FORTRAN Subprograms Used in Astronomy</p>	<p style="text-align: center;">J. A. Ball</p> <p style="text-align: center;">16 July 1969</p>
<p>Prepared under Electronic Systems Division Contract AF 19(628)-5167 by</p> <p style="text-align: center;">Lincoln Laboratory</p> <p style="text-align: center;">MASSACHUSETTS INSTITUTE OF TECHNOLOGY</p> <p style="text-align: center;">Lexington, Massachusetts</p>	

AD0691816

MASSACHUSETTS INSTITUTE OF TECHNOLOGY
LINCOLN LABORATORY

SOME FORTRAN SUBPROGRAMS USED IN ASTRONOMY

JOHN A. BALL

Group 21

TECHNICAL NOTE 1969-42

16 JULY 1969

This document has been approved for public release and sale;
its distribution is unlimited.

LEXINGTON

MASSACHUSETTS

The work reported in this document was performed at Lincoln Laboratory, a center for research operated by Massachusetts Institute of Technology, with the support of the Department of the Air Force under Contract AF 19(628)-5167.

This report may be reproduced to satisfy needs of U. S. Government agencies.

Abstract

This note is a description of the subprograms DOP, which calculates the Doppler velocity of an earth-bound observer, JULDA which calculates the Julian day-number, MOVE which calculates precession, COORD which performs coordinate transformations, and the GRM series of subprograms which deal with Doppler velocities in terms of a standard galactic rotation model. These subprograms are written in basic FORTRAN and should be useable on a wide variety of computers.

Accepted for the Air Force
Franklin C. Hudson
Chief, Lincoln Laboratory Office

SUBROUTINE DOP (RAHRS,RAMIN,ASEC,DDEC,DMIN,DSEC,NYR,NDAY,
2 NHUT,NMUT,NSUT,ALAT,OLONG,ELEV, XLST,VSUN,VMOON,VOBS,V1)

C
C DOP CALCULATES THE VELOCITY COMPONENT OF THE OBSERVER WITH RESPECT
C TO THE LOCAL STANDARD OF REST AS PROJECTED ONTO A LINE SPECIFIED BY THE RIGHT
C ASCENSION AND DECLINATION (RAHRS, RAMIN, ASEC, DDEC, DMIN, DSEC) EPOCH OF
C DATE, FOR A TIME SPECIFIED AS FOLLOWS+ NYR = LAST TWO DIGITS OF THE YEAR
C (FOR 19XX A.D.), NDAY = DAY NUMBER (GMT), NHUT, NMUT, NSUT = HRS, MIN, SEC
C (GMT). THE LOCATION OF THE OBSERVER IS SPECIFIED BY THE LATITUDE (ALAT),
C LONGITUDE (OLONG) (GEODETIC) (IN DEGREES) AND ELEVATION (ELEV) (IN METERS)
C ABOVE MEAN SEA LEVEL. THE SUBROUTINE OUTPUTS THE LOCAL MEAN SIDEREAL TIME
C (XLST IN DAYS), THE COMPONENT OF THE SUN'S MOTION WITH RESPECT TO THE LOCAL
C STANDARD OF REST AS PROJECTED ONTO THE LINE OF SIGHT TO THE SOURCE (VSUN IN
C KM/SEC) AS WELL AS THE TOTAL VELOCITY COMPONENT V1 (KM/SEC). POSITIVE
C VELOCITY CORRESPONDS TO INCREASING DISTANCE BETWEEN SOURCE AND OBSERVER.
C

C THIS VERSION OF DOP TAKES INTO ACCOUNT COMPONENTS OF THE OBSERVER'S
C MOTION DUE TO THE ROTATION OF THE EARTH, THE REVOLUTION OF THE EARTH-MOON
C BARYCENTER ABOUT THE SUN, AND THE MOTION OF THE EARTH'S CENTER ABOUT THE
C EARTH-MOON BARYCENTER. THE PERTURBATIONS OF THE EARTH'S ORBIT DUE TO THE
C PLANETS ARE NEGLECTED. THE ABSOLUTE PRECISION OF THIS VERSION OF DOP IS
C ABOUT 0.004 KM/SEC, BUT SINCE THE DOMINANT ERROR TERM IS SLOWLY VARYING, THE
C RELATIVE ERROR WILL BE CONSIDERABLY LESS FOR TIMES UP TO A WEEK OR SO.
C

C REFERENCES+ MCRAE, D. A., WESTERHOUT, G., TABLE FOR THE REDUCTION OF
C VELOCITIES TO THE LOCAL STANDARD OF REST, THE OBSERVATORY,
C LUND, SWEDEN, 1956.
C SMART, W. M., TEXT-BOOK ON SPHERICAL ASTRONOMY, CAMBRIDGE
C UNIV. PRESS, 1962.
C THE AMERICAN EPHEMERIS AND NAUTICAL ALMANAC
C THE SUPPLEMENT TO THE ABOVE
C

C VERSION OF JUNE 1969
C
C
C

```

C
C THE FOLLOWING CALCULATIONS DEAL WITH THE SUN'S MOTION WITH RESPECT TO THE
C LOCAL STANDARD OF REST.
C THE VELOCITY OF THE SUN WITH RESPECT TO THE LOCAL STANDARD OF REST IS THE
C CONVENTIONAL VALUE OF 20.0 KM/SEC TOWARD RA = 18 HRS, DEC = 30 DEG (1960).
  AAA=18.0*3.1415926535/12.0
  DD=30.0*3.1415926535/180.0
C MOVE PRECESSES THIS DIRECTION TO DATE
  CALL MOVE (1900,1900+NYR,1,NJAY,AAA,DD, DELA,DELOD,DC) MOVE
  AAA=AAA+DELA
  DD=DD+DELOD
C THIS VELOCITY IS CONVERTED TO CARTESIAN COMPONENTS
  XO=20.0*COSF(AAA)*COSF(DD)
  YO=20.0*SINF(AAA)*COSF(DD)
  ZO=20.0*SINF(DD)
C RA1 IS THE RIGHT ASCENSION (REVS=DAYS)
  RA1=(RAHRS+RAMIN/60.0+RASEC/3600.0)/24.0
C RA IS THE RIGHT ASCENSION (RADIAN)
  RA=2.0*3.1415926535*RA1
C DEC IS THE DECLINATION (RADIAN)
  DEC=3.1415926535*(DDEG+SIGNF(DMIN/60.0+DSEC/3600.0,DDFG))/180.0
C CC, CS, AND S ARE THE DIRECTION COSINES CORRESPONDING TO RA AND DEC
  CC=COSF(DEC)*COSF(RA)
  CS=COSF(DEC)*SINF(RA)
  S=SINF(DEC)
C VSUN IS THE PROJECTION ONTO THE LINE OF SIGHT TO THE STAR OF THE SUN'S
C MOTION WITH RESPECT TO THE LOCAL STANDARD OF REST (KM/SEC)
  VSUN=-XO*CC-YO*CS-ZO*S
C
C COORDINATES OF THE OBSERVER, LATITUDE (RADIAN), AND LONGITUDE (REVS=DAYS)
  CAT=ALAT*3.1415926535/180.0
  WLONG=OLONG/360.0
C

```

```

C
C THE FOLLOWING CALCULATIONS DEAL WITH THE TIME
C THE EPOCH IS 1900 JANUARY 0.5 UT = JULIAN DAY 2415020.0
C DU IS THE TIME FROM THE EPOCH TO JAN 0.0 OF THE CURRENT YEAR (DAYS)
  DU=(JULDA(1900+NYR)-2415020)-0.5
C TU IS DU CONVERTED TO JULIAN CENTURIES
  TU=DU/36525.0
C UTDA IS THE GMT FROM JAN 0.0 TO THE PRESENT (DAYS)
  UTDA=NDAY+NHUT/24.0+NMUT/1440.0+NSUT/86400.0
C SMD (SMALL D) IS THE TIME FROM THE EPOCH TO THE PRESENT (DAYS)
  SMD=DU+UTDA
C T IS SMD CONVERTED TO JULIAN CENTURIES
  T=SMD/36525.0
C START IS THE GREENWICH MEAN SIDEREAL TIME ON JAN 0.0 (DAYS)
C (THE EXTRA 129.1794 SECS CORRESPONDS TO THE 0.7 CENTURY SUBTRACTED FROM TU.
C THE PRECISION IS THEREBY IMPROVED.)
  START=(6.0+38.0/60.0+(45.836+129.1794+8640184.542*(TU-0.7)+0.0929
    2 *TU**2)/3600.0)/24.0
C C1 IS THE CONVERSION FACTOR FROM SOLAR TIME TO SIDEREAL TIME
  C1=(23.0+56.0/60.0+4.09054/3600.0)/24.0
C GST IS THE GREENWICH MEAN SIDEREAL TIME (DAYS)
  GST=START+UTDA/C1
C XLST IS THE LOCAL MEAN SIDEREAL TIME (FROM JAN 0) (DAYS)
  XLST=GST-WLONG
  XLST=XLST-IFIX(XLST)
C

```



```

C
C THE FOLLOWING CALCULATIONS DEAL WITH THE OBSERVER'S MOTION WITH
C RESPECT TO THE EARTH'S CENTER.
C REDUCTION OF GEODETIC LATITUDE TO GEOCENTRIC LATITUDE (ARCSSECONDS)
  DLAT=-((11.0*60.0+32.7430)*SINF(2.0*CAT)+1.1633*SINF(4.0*CAT)
  2 -0.0026*SINF(6.0*CAT))
C CONVERT CAT TO GEOCENTRIC LATITUDE (RADIANS)
  CAT=CAT+DLAT*3.1415926535/3600.0/180.0
C RHO IS THE RADIUS VECTOR FROM THE EARTH'S CENTER TO THE OBSERVER (METERS)
  RHO=-6378160.0*(0.998527073+0.001676438*COSF(2.0*CAT)-0.000003519
  2 *COSF(4.0*CAT)+0.000000008*COSF(6.0*CAT))+ELEV
C AND VRHO IS THE CORRESPONDING CIRCULAR VELOCITY (METERS/SIDEREAL DAY)
  VRHO=2.0*3.1415926535*RHO
C CONVERTED TO KILOMETERS/SEC
  VRHO=VRHO/24.0E3/3600.0*C1
C VOBS IS THE PROJECTION ONTO THE LINE OF SIGHT TO THE STAP OF THE VELOCITY
C OF THE OBSERVER WITH RESPECT TO THE EARTH'S CENTER (KM/SEC)
  VOBS=VRHO*COSF(CAT)*COSF(DEC)*SINF(2.0*3.1415926535*(XLST-RA1))
C

```

```

C
C THE FOLLOWING CALCULATIONS DEAL WITH THE EARTH'S ORBIT ABOUT THE SUN
C AM IS THE MEAN ANOMALY (OF THE EARTH'S ORBIT) (RADIAN)
  AM=(358.47583+0.9856002670*SMD-0.000150*T**2-0.000003*T**3)
  2 *3.1415926535/180.0
C PI IS THE MEAN LONGITUDE OF PERHELION (RADIAN)
  PI=(101.22083+0.0000470684*SMD+0.000453*T**2+0.000003*T**3)
  2 *3.1415926535/180.0
C E IS THE ECCENTRICITY OF THE ORBIT (DIMENSIONLESS)
  E=0.01675104-0.00004180*T-0.000000126*T**2
C AI IS THE MEAN OBLIQUITY OF THE ECLIPTIC (RADIAN)
  AI=(23.452294-0.0130125*T-0.00000164*T**2+0.000000503*T**3)
  2 *3.1415926535/180.0
C VS IS THE TRUE ANOMALY (APPROXIMATE FORMULA) (RADIAN)
C (EQUATION OF THE CENTER)
  VS=AM+(2.0*E-0.25*E**3)*SINF(AM)+1.25*E**2*SINF(2.0*AM)+
  2 13.0/12.0*E**3*SINF(3.0*AM)
C XLAM IS THE TRUE LONGITUDE OF THE EARTH AS SEEN FROM THE SUN (RADIAN)
  XLAM=PI+VS
C ALAM IS THE TRUE LONGITUDE OF THE SUN AS SEEN FROM THE EARTH (RADIAN)
  ALAM=XLAM+3.1415926535
C BETA IS THE LATITUDE OF THE STAR (RADIAN)
C ALONG IS THE LONGITUDE OF THE STAR (RADIAN)
  CALL COORD (0.0,0.0,-3.1415926535/2.0,3.1415926535/2.0-AI,
  2 RA,DEC, ALONG,BETA)
C AA IS THE SEMI-MAJOR AXIS OF THE EARTH'S ORBIT (KM)
  AA=149598500.0
C AN IS THE MEAN ANGULAR RATE OF THE EARTH ABOUT THE SUN (RADIAN/DAY)
  AN=2.0*3.1415926535/365.2564
C HOP IS H/P FROM SMART = THE COMPONENT OF THE EARTH'S VELOCITY PERPENDICULAR
C TO THE RADIUS VECTOR (KM/DAY)
  HOP=AN*AA/SGRTF(1.0-E**2)
C CONVERTED TO KM/SEC
  HCP=HOP/86400.0
C V IS THE PROJECTION ONTO THE LINE OF SIGHT TO THE STAR OF THE VELOCITY
C OF THE EARTH-MOON BARYCENTER WITH RESPECT TO THE SUN (KM/SEC)
  V=-HCP*COSF(BETA)*(SINF(ALAM-ALONG)-E*SINF(PI-ALONG))
C

```

```

C
C THE FOLLOWING CALCULATIONS DEAL WITH THE MOON'S ORBIT AROUND THE
C EARTH-MOON BARYCENTER
C OMGA (OMEGA) IS THE LONGITUDE OF THE MEAN ASCENDING NODE OF THE LUNAR ORBIT
C (DEGREES)
  OMGA =259.183275-0.0529539222*SMD+0.002078*T**2+0.000002*T**3
C OMGAR IS OMGA IN RADIANS
  OMGAR=OMGA*3.1415926535/180.0
C AMON IS OMGA PLUS THE MEAN LUNAR LONGITUDE OF THE MOON (DEGREES)
C (SHOULD BE 13.1763965268)
  AMON=270.434164+13.176396527 *SMD-0.001133*T**2+0.0000019*T**3
C GAMP (GAMMA-PRIME) IS OMGA PLUS THE LUNAR LONGITUDE OF LUNAR PERIGEE (DEGREES)
  GAMP=334.329556+0.1114040803*SMD-0.010325*T**2-0.000012*T**3
C PIM IS THE MEAN LUNAR LONGITUDE OF LUNAR PERIGEE (TO RADIANS)
  PIM=(GAMP-OMGA)*3.1415926535/180.0
C EM IS THE ECCENTRICITY OF THE LUNAR ORBIT
  EM=0.054900489
C OLAMM IS THE MEAN LUNAR LONGITUDE OF THE MOON (TO RADIANS)
  OLAMM=(AMON-OMGA)*3.1415926535/180.0
C AIM IS THE INCLINATION OF THE LUNAR ORBIT TO THE ECLIPTIC (RADIANS)
  AIM=5.1453964*3.1415926535/180.0
C AMM IS THE APPROXIMATE MEAN ANOMALY (RADIANS)
C (IT IS APPROXIMATE BECAUSE PIM SHOULD BE THE TRUE RATHER THAN THE MEAN LUNAR
C LONGITUDE OF LUNAR PERIGEE)
  AMM=OLAMM-PIM
C VSM IS THE TRUE ANOMALY (APPROXIMATE FORMULA) (RADIANS)
C (EQUATION OF THE CENTER)
  VSM=AMM+(2.0*EM-0.25*EM**3)*SINF(AMM)+1.25*EM**2*SINF(2.0*AMM)
  2 +13.0/12.0*EM**3*SINF(3.0*AMM)
C ALAMM IS THE TRUE LUNAR LONGITUDE OF THE MOON (RADIANS)
  ALAMM=PIM+VSM
C ANM IS THE MEAN ANGULAR RATE OF THE LUNAR ROTATION (RADIANS/DAY)
  ANM=2.0*3.1415926535/27.321661
C AAM IS THE SEMI-MAJOR AXIS OF THE LUNAR ORBIT (KILOMETERS)
  AAM=60.2665*6378.388
C BETAM IS THE LUNAR LATITUDE OF THE STAR (RADIANS)
C ALGM IS THE LUNAR LONGITUDE OF THE STAR (RADIANS)
  CALL COORD (OMGAR,0.0,OMGAR-3.1415926535/2.0,3.1415926535/2.0-AIM,
  2 ALONG,BETA, ALGM,BETAM)
C HOPM IS H/P FROM SMART = THE COMPONENT OF THE LUNAR VELOCITY PERPENDICULAR
C TO THE RADIUS VECTOR (KM/DAY)
  HOPM=ANM*AAM/SQRTF(1.0-EM**2)
C CONVERTED TO KM/SEC
  HOPM=HOPM/86400.0
C VMON IS THE PROJECTION ONTO THE LINE OF SIGHT TO THE STAR OF THE VELOCITY
C OF THE EARTH'S CENTER WITH RESPECT TO THE EARTH-MOON BARYCENTER (KM/SEC)
C (THE 81.30 IS THE RATIO OF THE EARTH'S MASS TO THE MOON'S MASS)
  VMON=-HOPM/81.30*COSF(BETAM)*(SINF(ALAMM-ALGM)-EM*SINF(PIM-ALGM))
C

```

```
C      V1=V+VSUN+VMON+VOBS
      RETURN
C
C THIS PROGRAM OMITTS THE PLANETARY PERTURBATIONS ON THE EARTH'S ORBIT. THESE
C AMOUNT TO ABOUT 0.003 KM/SEC AND ARE THOUGHT TO BE THE LARGEST CONTRIBUTION
C TO THE ERROR IN THE VELOCITY.
C
      END
```

```
      FUNCTION JULDA(NYR)
C     THIS FUNCTION COMPUTES THE JULIAN DAY NUMBER AT 12 HRS UT ON JANUARY 0
C     OF THE YEAR NYR (GREGORIAN CALENDAR).  JULDA IS AN INTEGER BECAUSE OF THIS
C     DEFINITION.  FOR EXAMPLE, JULDA = 2439856 FOR NYR = 1969.
C
      NYRM1=NYR-1
      IC=NYRM1/100
      JULDA=1721425+365*NYRM1+NYRM1/4-IC+IC/4
      RETURN
      END
```

```

SUBROUTINE COORD (A0,B0,AP,BP,A1,B1, A2,B2)
C
C THIS SUBROUTINE CONVERTS THE LONGITUDE-LIKE (A1) AND LATITUDE-LIKE (B1)
C COORDINATES OF A POINT ON A SPHERE INTO THE CORRESPONDING COORDINATES (A2,
C B2) IN A DIFFERENT COORDINATE SYSTEM THAT IS SPECIFIED BY THE COORDINATES
C OF ITS ORIGIN (A0, B0) AND ITS NORTH POLE (AP, BP) IN THE ORIGINAL COORDINATE
C SYSTEM. THE RANGE OF A2 WILL BE FROM -PI TO PI.
C
C ALL ARGUMENTS ARE IN RADIANs.
C
C EXAMPLES OF USE
C   PI = 3.1415926535
C   PI02 = PI/2.0
C
C EXAMPLE I--TO CALCULATE AZIMUTH AND ELEVATION FROM HOUR ANGLE AND DECLINATION
C   CALL COORD (PI,PI02-LATITUDE,0.0,LATITUDE,HOUR ANGLE,DECLINATION,
C   2 AZIMUTH,ELEVATION)
C THEN IF AZIMUTH IS DESIRED IN THE RANGE 0 TO PI SET
C   AZIMUTH = AZIMUTH + (PI - SIGNF(PI,AZIMUTH))
C
C EXAMPLE II--TO CALCULATE HOUR ANGLE AND DECLINATION FROM AZIMUTH AND
C ELEVATION
C   CALL COORD (PI,PI02-LATITUDE,0.0,LATITUDE,AZIMUTH,ELEVATION,
C   2 HOUR ANGLE,DECLINATION)
C
C EXAMPLE III--TO CALCULATE LI AND BI FROM RIGHT ASCENSION AND DECLINATION
C (EPOCH 1900.0)
C   AP = (12.0+40.0/60.0)*PI/12.0
C (I.E. 12 HOURS 40 MINUTES CONVERTED TO RADIANs)
C   BP = 28.0*PI/180.0
C   AO = (18.0+40.0/60.0)*PI/12.0
C   BO = 0.0
C (REFER TO KRAUS, P., RADIO ASTRONOMY, MCGRAW HILL, NEW YORK, 1966. BUT FOR
C FURTHER REFINEMENTS, SEE ALSO ALLEN, C. W., ASTROPHYSICAL QUANTITIES,
C ATHLONE PRESS, LONDON, 1963.)
C   CALL COORD (AO,BO,AP,BP,RIGHT ASCENSION,DECLINATION,LI,BI)
C

```

```

C
C EXAMPLE IV--TO CALCULATE RIGHT ASCENSION AND DECLINATION (EPOCH 1900.0) FROM
C LI AND BI
C IN GENERAL, WHENEVER WE KNOW THE FORWARD TRANSFORMATION (EXAMPLE III ABOVE)
C WE MAY DO THE REVERSE TRANSFORMATION WITH AT MOST TWO EXTRA PRELIMINARY CALLS
C TO COORD TO CALCULATE THE COORDINATES IN SYSTEM 2 OF THE POLAR ORIGIN IN
C SYSTEM 1. BUT OFTEN IT IS POSSIBLE TO GET THESE NEEDED COORDINATES BY
C INSPECTION. FOR EXAMPLE, BP WILL REMAIN THE SAME FOR THE FORWARD AND REVERSE
C TRANSFORMATIONS. FOR THIS EXAMPLE WE SEE BY INSPECTION THAT
C   APP = 6.0*PI/12.0
C   BPP = 28.0*PI/180.0
C (THE SECOND P REPRESENTS PRIME.) AND WE MAY CALCULATE AOP AND BOP FROM
C   CALL COORD (AO,BO,AP,BP,0.0,0.0,AOP,BOP)
C WHERE THE AO, ETC. ARE FROM EXAMPLE III. THEN THE ACTUAL CONVERSION IS
C   CALL COORD (AOP,BOP,APP,BPP,LI,BI,RIGHT ASCENSION,DECLINATION)
C
C EXAMPLE V--TO CALCULATE LII AND BII FROM RIGHT ASCENSION AND DECLINATION
C (EPOCH 1950.0)
C   AP = (12.0+49.0/60.0)*PI/12.0
C   BP = 27.4*PI/180.0
C   AO = (17.0+42.4/60.0)*PI/12.0
C   BO = -(28.0+55.0/60.0)*PI/180.0
C   CALL COORD (AO,BO,AP,BP,RIGHT ASCENSION,DECLINATION,LII,BII)
C
C EXAMPLE VI--TO CALCULATE RIGHT ASCENSION AND DECLINATION (EPOCH 1950.0) FROM
C LII AND BII
C FIRST CALCULATE APP AND BPP FROM
C   CALL COORD (AO,BO,AP,BP,0.0,PI/2,APP,BPP)
C THEN CALCULATE AOP AND BOP FROM
C   CALL COORD (AO,BO,AP,BP,0.0,0.0,AOP,BOP)
C WHERE THE AO, ETC. ARE FROM EXAMPLE V. THEN THE ACTUAL CONVERSION IS
C   CALL COORD (AOP,BOP,APP,BPP,LII,BII,RIGHT ASCENSION,DECLINATION)
C

```

C
C EXAMPLE VII--TO CALCULATE (ECLIPTIC) LATITUDE AND LONGITUDE FROM RIGHT
C ASCENSION AND DECLINATION
C EPS IS THE OBLIQUITY OF THE ECLIPTIC WHICH IS ABOUT 23.443 DEGREES, BUT IT
C DEPENDS ON THE EPOCH. SEE THE AMERICAN EPHEMERIS AND NAUTICAL ALMANAC.
C EPS=23.443*PI/180.0
C CALL COORD (0.0,0.0,-PI/2,PI/2-EPS,RIGHT ASCENSION,DECLINATION,
C 2 LATITUDE,LONGITUDE)
C
C
C THE NOTATION USES S OR C FOR SINE OR COSINE OF THE CORRESPONDING VARIABLE,
C FOR EXAMPLE, SBO = SIN(BO), ETC.
C
C NOTE THAT THE INPUT PARAMETERS ARE PARTIALLY REDUNDANT. FOR EXAMPLE, IF
C AP, BP, AND AO ARE SPECIFIED, THEN THERE ARE ONLY TWO DISCRETE VALUES
C POSSIBLE FOR BO (EXCEPT FOR A FEW DEGENERATE SPECIAL CASES). SEE BELOW FOR
C WHAT TO DO IF IT IS NECESSARY TO PRECALCULATE AO AND BO.
C
C IF, INSTEAD OF AO AND BO, THE LONGITUDE OF THE ASCENDING NODE IS KNOWN IN
C BOTH THE OLD (AN1) AND NEW (AN2) COORDINATE SYSTEMS, THEN AO AND BO MAY BE
C CALCULATED BY A PRELIMINARY CALL TO COORD
C CALL COORD (0.0,0.0,AN1-AP,BP,-AN2,0.0,AO,BO)
C THEN THIS AO AND BO MAY BE USED FOR A SERIES OF ORDINARY CALLS TO COORD AS
C DESCRIBED ABOVE.
C
C IF AP, BP, AND AO ARE KNOWN, THEN THE TWO POSSIBLE VALUES OF BO MAY BE
C CALCULATED FROM
C
$$SBO = \frac{(SBP \pm 2.0 * CBP ** 2 * CAPAO * \text{SQRT}(1.0 + CAPAO ** 2))}{2 * (SBP ** 2 + (CBP * CAPAO) ** 2)}$$
C WHERE CAPAO = COS(AO-AP) AND THE OTHER NOTATION IS EXPLAINED ABOVE, AND
C WHERE THE \$ IS TO BE REPLACED BY + AND -.
C
C IF AP, BP, AND BO ARE KNOWN, THEN THE TWO POSSIBLE VALUES OF AO MAY BE
C CALCULATED FROM
C
$$CAPAO = (1.0 - SBO * SBP) / (CBO * CBP)$$
C BOTH ANGLES WITH THIS COSINE ARE POSSIBLE.
C


```

C      SB0=SINF(B0)
      CB0=COSF(B0)
      SBP=SINF(BP)
      CBP=COSF(BP)
      SB1=SINF(B1)
      CB1=COSF(B1)
C
C      SB2=SBP*SB1+CBP*CB1*COSF(AP-A1)
      B2=ASINF(SB2)
C (NOTE B0 IS NOT NEEDED TO CALCULATE B2)
      CB2=COSF(B2)
C
C      SAA=SINF(AP-A1)*CB1/CB2
      CAA=(SB1-SB2*SBP)/(CB2*CBP)
C
C      CBB=SB0/CBP
      SBB=SINF(AP-A0)*CB0
C
C      TA202=(1.0-CAA*CBB-SAA*SBB)/(SAA*CBB-CAA*SBB)
      A2=2.0*ATANF(TA202)
C
      RETURN
      END

```

```

SUBROUTINE MOVE (NYRI,NYRF,MO,NDA,RA,D, DELR,DELD,DC)
C
C MOVE CALCULATES THE CORRECTION (DELR) IN RIGHT ASCENSION (RA) AND THE
C CORRECTION (DELD) IN DECLINATION (D) (ALL IN RADIANS) TO BE ADDED TO THE
C MEAN COORDINATES FOR EPOCH NYRI (E.G. 1950) TO GIVE THE APPARENT POSITIONS
C OF A DATE SPECIFIED BY THE YEAR (NYRF, E.G. 1968), MONTH (MO, 1 TO 12), AND
C DAY (NDA). IF THE DAY-NUMBER IS KNOWN, USE IT FOR NDA AND SET MO = 1.
C MOVE ALSO CALCULATES THE EQUATION OF THE EQUINOXES (DC, IN MINUTES OF TIME)
C WHICH MAY BE ADDED TO THE MEAN SIDEREAL TIME TO GIVE THE APPARENT SIDEREAL
C TIME (AENA+469). DELR AND DELD CONTAIN CORRECTIONS FOR PRECESSION, ANNUAL
C ABERRATION, AND SOME TERMS OF NUTATION. IF RA AND D ARE FOR THE MEAN EPOCH
C (I.E. HALFWAY BETWEEN NYRI AND NYRF) THEN THE PRECISION OF DELR AND DELD IS
C ABOUT 2 ARCSECONDS (SEE NEGLECTED TERMS IN ESE-44). IF RA AND D ARE EITHER
C OF THE END POINTS OF THE INTERVAL, THEN THE PRECISION MAY BE SOMEWHAT WORSE.
C AENA = THE AMERICAN EPHEMERIS AND NAUTICAL ALMANAC (THE BLUE BOOK).
C ESE = THE EXPLANATORY SUPPLEMENT TO ABOVE (THE GREEN BOOK).
C
      SND=SINF(D)
      CSD=COSF(D)
      TND=SND/CSD
C
      CSR=COSF(RA)
      SNR=SINF(RA)
C
C AL IS AN APPROXIMATE DAY NUMBER (I.E. THE NUMBER OF DAYS SINCE JANUARY 0
C OF THE YEAR NYRF).
      AL=30*(MO-1)+NDA
C
C TO IS THE TIME FROM 1900 TO NYRI (CENTURIES)
      TO=FLOATF(NYRI-1900)/100.0
C T IS THE TIME FROM NYRI TO DATE (NYRF, MO, NDA) (CENTURIES)
C (365.2421988 IS THE NUMBER OF EPHEMERIS DAYS IN A TROPICAL YEAR)
      T=(FLOATF(NYRF-NYRI)+AL/365.2421988)/100.0
C ZETA0 IS A PRECESSIONAL ANGLE FROM ESE-29 (ARCSECONDS)
      ZETA0=(2304.250+1.396*T0)*T+0.302*T**2+0.018*T**3
C DITTO FOR Z
      Z=ZETA0+0.791*T**2
C AND THETA
      THETA=(2004.682-0.853*T0)*T-0.426*T**2-0.042*T**3
C AM AND AN ARE THE M AND N PRECESSIONAL NUMBERS (SEE AENA-50, 474) (RADIANS)
      AM=(ZETA0+Z)*4.848136811E-6
      AN=THETA*4.848136811E-6
C

```

```

C
C ALAM IS AN APPROXIMATE MEAN LONGITUDE FOR THE SUN (AFNA-50) (RADIAN)
  ALAM=(0.985647*AL+278.5)*0.0174532925
  SNL=SINF(ALAM)
  CSL=COSF(ALAM)
C DELR IS THE ANNUAL ABERRATION TERM IN RA (RADIAN) (ESE-47,4P)
C (0.91745051 = COS(OBLIQUITY OF ECLIPTIC))
C (-9.92413605E-5 = K = 20.47 ARCSECONDS = CONSTANT OF ABERRATION (ESE-4P))
  DELR=-9.92413605E-5*(SNL*SNR+0.91745051*CSL*CSR)/CSD
C PLUS PRECESSION TERMS (SEE AENA-50 AND ESE-38)
  2 +AM+AN*SNR*TND
C DELD IS DITTO ABOVE IN DECLINATION
  DELD=-9.92413605E-5*(SNL*CSR*SND-0.91745051*CSL*SNR*SND
C (0.39784993 = SIN(OBLIQUITY OF ECLIPTIC))
  2 +0.39784993*CSL*CSD) +AN*CSR
C
C THE FOLLOWING CALCULATES THE NUTATION (APPROXIMATELY) (ESE-41,45)
C OMEGA IS THE ANGLE OF THE FIRST TERM OF NUTATION (ESE-44) (APPROXIMATE
C FORMULA) (DEGREES)
  OMEGA=259.183275-1934.142*(TO+T)
C ARG IS OMEGA CONVERTED TO RADIAN
  ARG=OMEGA*0.0174532925
C DLONG IS THE NUTATION IN LONGITUDE (DELTA-PSI) (RADIAN)
  DLONG=-8.3597E-5*SINF(ARG)
C DOBLQ IS THE NUTATION IN OBLIQUITY (DELTA-EPSILON) (RADIAN)
  DOBLQ= 4.4678E-5*COSF(ARG)
C
C ADD NUTATION IN RA INTO DELR (ESE-43)
  DELR=DELR+DLONG*(0.91745051 +0.39784993 *SNR*TND)-CSR*TND*DOBLQ
C AND DEC.
  DELD=DELD+0.39784993 *CSR*DLONG+SNR*DOBLQ
C DC IS THE EQUATION OF THE EQUINOXES (MINUTES OF TIME) (ESE-43)
  DC=DLONG*210.264169
  RETURN
  END

```

```

SUBROUTINE GRM1 (ELII,VR, RS,RC,PSI)
DIMENSION RS(2),PSI(2)
C SUBROUTINE GRM1 (GALACTIC ROTATION MODEL - 1) ACCEPTS THE GALACTIC LONGITUDE
C (ELII IN DEGREES) AND THE DOPPLER VELOCITY WITH RESPECT TO THE LOCAL STANDARD
C OF REST (VR IN KM/SEC) AND RETURNS THE DISTANCE FROM THE SUN (RS), THE
C DISTANCE FROM THE GALACTIC CENTER (RC) (BOTH IN KPC), AND THE GALACTOCENTRIC
C LONGITUDE (PSI IN DEGREES). THE RANGE OF PSI WILL BE -180 TO +180 DEGREES.
C FOR MANY VALUES OF ELII AND VR, THERE WILL BE A DISTANCE AMBIGUITY AND THE
C TWO POSSIBLE DISTANCES WILL BE RETURNED IN RS(1) AND RS(2) AND THE ASSOCIATED
C ANGLES IN PSI(1) AND PSI(2). IF RS(1) IS NEGATIVE OR ZERO, THEN THERE IS
C ONLY ONE SOLUTION AND IT IS IN RS(2) AND PSI(2). IF BOTH RS(1) AND RS(2)
C ARE NEGATIVE OR ZERO, THEN THERE ARE NO SOLUTIONS TO THE EQUATIONS, I.E. THE
C GIVEN VR IS IMPOSSIBLE FOR THIS ELII.
C SEE SUBROUTINE GRM3 FOR FURTHER DETAILS.
C
PI=3.1415926535
TPI=2.0*PI
C ELIIR IS ELII CONVERTED TO RADIANS
ELIIR=ELII*PI/180.0
C SLII IS SIN(ELII)
SLII=SINF(ELIIR)
C RO IS THE SUN'S DISTANCE FROM THE GALACTIC CENTER (KPC)
RO=10.0
C OMGAO IS THE ANGULAR ROTATION VELOCITY AT THE SUN (KM/SEC/KPC)
OMGAO=25.0
C OMGA IS THE ANGULAR ROTATION VELOCITY AT THE SOURCE (KM/SEC/KPC)
OMGA=VR/(RO*SLII)+OMGAO
C GRM3 CALCULATES RC (KPC)
CALL GRM3 (OMGA, RC)
C H IS THE LENGTH OF THE PERPENDICULAR DROPPED FROM THE GALACTIC CENTER TO THE
C LINE OF SIGHT (KPC)
H=RO*SLII
C GO TO 50 IF THERE IS NO SOLUTION
IF (ABSF(H)-RC) 20,20,50
C DELT IS THE ANGLE BETWEEN THE SUN AND THE GALACTIC CENTER AS SEEN FROM THE
C SOURCE (RADIANS)
20 DELT=ASINF(H/RC)
C PSI1 AND PSI2 ARE THE TWO SOLUTIONS FOR PSI (RADIANS)
PSI1=ELIIR-DELT
PSI2=ELIIR+DELT+PI
C CONVERT THESE TO DEGREES, SET THE RANGE TO -180 TO 180, AND PUT IN PSI
PSI(1)=PSI1*180.0/PI-360.0*IFIX(PSI1/TPI+SIGNF(0.5,PSI1))
PSI(2)=PSI2*180.0/PI-360.0*IFIX(PSI2/TPI+SIGNF(0.5,PSI2))
C THEN SET RS(1) AND RS(2) (KPC)
RS(1)=-RC*SINF(PSI1)/SLII
RS(2)=-RC*SINF(PSI2)/SLII
RETURN
C
C GO TO 50 FOR NO SOLUTIONS TO THE EQUATIONS (ABSF(H/RC) .GT. 1)
50 RS(1)=0.0
RS(2)=0.0
PSI(1)=0.0
PSI(2)=0.0
RETURN
END

```

```

SUBROUTINE GRM2 (ELII,RS, VR,RC,PSI)
C SUBROUTINE GRM2 (GALACTIC ROTATION MODEL - 2) ACCEPTS THE GALACTIC LONGITUDE
C (ELII IN DEGREES) AND THE DISTANCE FROM THE SUN (RS IN KPC) AND RETURNS THE
C DOPPLER VELOCITY WITH RESPECT TO THE LOCAL STANDARD OF REST (VR IN KM/SEC),
C THE DISTANCE FROM THE GALACTIC CENTER (RC IN KPC), AND THE GALACTOCENTRIC
C LONGITUDE (PSI IN DEGREES).
C SEE SUBROUTINE GRM4 FOR FURTHER DETAILS.
C
PI=3.1415926535
C ELIIR IS ELII CONVERTED TO RADIANS
ELIIR=ELII*PI/180.0
C RO IS THE SUN'S DISTANCE FROM THE GALACTIC CENTER (KPC)
RO=10.0
C THEN RC IS JUST
RC=SQRTF(RS**2+RO**2-2.0*RS*RO*COSF(ELIIR))
C OMGAS (FROM GRM4) IS THE ANGULAR ROTATION VELOCITY AT THE SOURCE (KM/SEC/KPC)
CALL GRM4 (RC, OMGAS) GRM4
C OMGASO IS THE ANGULAR ROTATION VELOCITY AT THE SUN (KM/SEC/KPC)
OMGASO=25.0
C AND THEN VR IS JUST
VR=(OMGAS-OMGASO)*RO*SINF(ELIIR)
C A IS THE ANGLE BETWEEN THE SUN AND THE SOURCE AS SEEN FROM THE GALACTIC
C CENTER (RADIANS)
C SNA IS SIN(A) (DIMENSIONLESS)
SNA=RS*SINF(ELIIR)/RC
C CSA IS COS(A) (DIMENSIONLESS)
CSA=(RO**2+RC**2-RS**2)/(2.0*RO*RC)
C THEN A IS JUST
A=2.0*ATANF(SNA/(1.0+CSA))
C AND PSI (CONVERTED TO DEGREES) IS JUST
PSI=-A*180.0/PI
RETURN
END

```

```

SUBROUTINE GRM3 (OMGA, RC)
C SUBROUTINE GRM3 (GALACTIC ROTATION MODEL = 3) ACCEPTS THE ANGULAR ROTATION
C VELOCITY (OMGA IN KM/SEC/KPC) AND RETURNS THE DISTANCE FROM THE GALACTIC
C CENTER (RC IN KPC), BASED ON THE MODEL OF THE GALAXY SUGGESTED BY SCHMIDT
C (1965). SEE SUBROUTINE GRM4 FOR FURTHER DETAILS.
C
C IF OMGA .LE. 0 THEN RC = 0
  IF (OMGA) 10,10,20
10 RC=0.0
  RETURN
C
C UNFORTUNATELY THERE IS NO WAY TO INVERT SCHMIDT'S EQUATIONS TO OBTAIN RC
C DIRECTLY FROM OMGA. THE FOLLOWING ITERATION PROCEDURE SEEMS TO WORK AS WELL
C AS ANY. IT IS BASED ON THE APPROXIMATION THAT THE CIRCULAR VELOCITY
C (=OMGA*RC) IS A SLOWLY VARYING FUNCTION OF RC. THE NUMBER 200 IS A FIRST
C GUESS AT THIS VELOCITY.
20 RC=200.0/OMGA
  DO 30 I=1,5
    CALL GRM4 (RC, OMGAT)
    B=(2.0+10.0/(10.0+RC))/3.0
    RC=RC+RC*B*(OMGAT-OMGA)/OMGA
30 CONTINUE
  RETURN
END

```

```

      SUBROUTINE GRM4 (RC, OMGA)
C   SUBROUTINE GRM4 (GALACTIC ROTATION MODEL - 4) ACCEPTS THE DISTANCE FROM THE
C   GALACTIC CENTER (RC IN KPC) AND RETURNS THE ANGULAR ROTATION VELOCITY (OMGA
C   IN KM/SEC/KPC).  BASED ON THE MODEL OF THE GALAXY SUGGESTED BY MAARTEN
C   SCHMIDT, CHAP. 22 IN GALACTIC STRUCTURE, ED. BY A. BLAAUW AND M. SCHMIDT,
C   UNIV. OF CHICAGO PRESS, 1965.
C
C   IF RC .LE. 0 THEN OMGA = 0
      IF (RC) 10,10,20
10    OMGA=0.0
      RETURN
C
20    IF (RC-10.0) 40,30,50
30    OMGA=25.0
      RETURN
C
C   THE FOLLOWING CALCULATIONS ARE FOR RC .LT. 10 KPC
C   VC IS THE CIRCULAR VELOCITY (KM/SEC)
40    VC=SQRTF(30000.0/RC+10120.2*RC-41.722*RC**3)
C   THEN OMGA IS JUST
      OMGA=VC/RC
      RETURN
C
C   THE FOLLOWING CALCULATIONS ARE FOR RC .GT. 10 KPC
50    VC=SQRTF(851611.6/RC-2148585.1/RC**2)
      OMGA=VC/RC
      RETURN
      END

```

DOCUMENT CONTROL DATA - R&D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall report is classified)

1. ORIGINATING ACTIVITY <i>(Corporate author)</i>		2a. REPORT SECURITY CLASSIFICATION Unclassified	
Lincoln Laboratory, M. I. T.		2b. GROUP None	
3. REPORT TITLE Some FORTRAN Subprograms Used in Astronomy			
4. DESCRIPTIVE NOTES <i>(Type of report and inclusive dates)</i> Technical Note			
5. AUTHOR(S) <i>(Last name, first name, initial)</i> Ball, John A.			
6. REPORT DATE 16 July 1969		7a. TOTAL NO. OF PAGES 24	7b. NO. OF REFS 7
8a. CONTRACT OR GRANT NO. AF 19(628)-5167		9a. ORIGINATOR'S REPORT NUMBER(S) Technical Note 1969-42	
b. PROJECT NO. 649L		9b. OTHER REPORT NO(S) <i>(Any other numbers that may be assigned this report)</i> ESD-TR-69-206	
c.			
d.			
10. AVAILABILITY/LIMITATION NOTICES This document has been approved for public release and sale; its distribution is unlimited.			
11. SUPPLEMENTARY NOTES None		12. SPONSORING MILITARY ACTIVITY Air Force Systems Command, USAF	
13. ABSTRACT This note is a description of the subprograms DOP, which calculates the Doppler velocity of an earth-bound observer, JULDA which calculates the Julian day-number, MOVE which calculates precession, COORD which performs coordinate transformations, and the GRM series of subprograms which deal with Doppler velocities in terms of a standard galactic rotation model. These subprograms are written in basic FORTRAN and should be useable on a wide variety of computers.			
14. KEY WORDS FORTRAN Doppler velocities computer programming galactic rotation model			